7次方程式を解く例 with Mathematica14.0

-詳しい解説は他のファイルをご覧ください-

§0 まとめ

In[3525]:=

```
ClearAll["`*"]
```

F35(「F35.nb」参照　)

f_1, f_2, f_3, g_1, g_2 を求めるプログラム(「F1F2F3.nb, G1G2.nb」参照. ↓を評価してください)

In[3527]:=

```
findF12G12[f_] := Module[{f35, factors, pos1, pos2, f12, d,R1,R2, f12factors,u,v,fp,fm,g1,g2,gcd1,gcd2},
Clear[a1,a2,a3,a4,a5,a6,a7];
  f35 = F35 /. AssociationThread[{a1, a2, a3, a4, a5, a6, a7} → Reverse@CoefficientList[f, x, 7]];
  factors = FactorList[f35];
  pos1 = Position[Exponent[factors⟦All, 1⟧, x], 14]⟦1, 1⟧;
  f12 = factors⟦pos1⟧⟦1⟧;
  d = Sqrt@Discriminant[f12, x];
  f12factors= FactorList[f12, Extension → d];
  pos2 = Flatten@Position[Exponent[f12factors⟦All, 1⟧, x], 7];
  {$f1,$f2}=(Expand[#/Coefficient[#, x, 7]] &) /@f12factors⟦pos2⟧⟦All, 1⟧;
  u=-(t1+t2)/14+(t1-t2)/(2Sqrt[-7]);
  v=-(t1+t2)/14-(t1-t2)/(2Sqrt[-7]);
 fp=($f1/.{x→u})+($f2/.{x→v});
 fm=Sqrt[-7](($f1/.{x→u})-($f2/.{x→v}));
g1=GroebnerBasis[{f/.{x→(t1+t2)/7},fp,fm},{t1},{t2}]⟦1⟧;
$g1=If[Coefficient[g1,t1,7]=!=0,Expand[#/Coefficient[#,t1,7]]&[g1]/.{t1→x},
R1=Resultant[f/.{x→(t1+t2)/7},fp/.{u→-(t1+t2)/14-Sqrt[-7](t1-t2)/14,v→-(t1+t2)/14+Sqrt[-7](t1-t2)/14},t2];
 R2=Resultant[f/.{x→(t1+t2)/7},fm/.{u→-(t1+t2)/14-Sqrt[-7](t1-t2)/14,v→-(t1+t2)/14+Sqrt[-7](t1-t2)/14},t2];
 PolynomialGCD[R1,R2,Extension→Automatic];
gcd1=PolynomialGCD[R1,R2,Extension→Automatic];
Expand[gcd1/Coefficient[gcd1,t1,7]]/.{t1→x}];
g2=GroebnerBasis[{f/.{x→(t1+t2)/7},fp,fm},{t2},{t1}]⟦1⟧;
$g2=If[Coefficient[g2,t2,7]=!=0,Expand[#/Coefficient[#,t2,7]]&[g2]/.{t2→x},
R1=Resultant[f/.{x→(t1+t2)/7},fp/.{u→-(t1+t2)/14-Sqrt[-7](t1-t2)/14,v→-(t1+t2)/14+Sqrt[-7](t1-t2)/14},t1];
 R2=Resultant[f/.{x→(t1+t2)/7},fm/.{u→-(t1+t2)/14-Sqrt[-7](t1-t2)/14,v→-(t1+t2)/14+Sqrt[-7](t1-t2)/14},t1];
 gcd2=PolynomialGCD[R1,R2,Extension→Automatic];
Expand[gcd2/Coefficient[gcd2,t2,7]]/.{t2→x}];
If[$g1===x^7,{$f1,$f2,$g1,$g2}={$f2,$f1,$g2,$g1}];
Return[{$f1,$f2,$g1,$g2}]]

(*findF3は$f1,$f2が必要なので, findF12G12が既に実行されている事が必要*)
findF3[f_]:=Module[{R1,R2,gcd3},
R1=Resultant[f,$f1/.{x→(t-x)/2},x];
R2=Resultant[f,$f2/.{x→(-t-x)/2},x];
gcd3=PolynomialGCD[R1,R2,Extension→Automatic];
$f3=(Expand[#/Coefficient[#,t,7]]&[gcd3])/.{t→x}]
```

r_1, r_2, r_4 を求める公式（「Formula1.nb」参照）

（b_i, c_i は 各々g_1, g_2 の（7 − i）次の係数）

「$b_3 = b_4 = b_5 = b_6 = 0$ でない時」に r_1, r_2, r_4 を求める公式

$$\begin{cases} R_1 = r_1^7 & (11) \\ R_1^3 + \left(\frac{b_3 b_4}{14} + b_7\right) R_1^2 + \left(-\frac{b_3^2 b_4^2}{9604} - \frac{b_3^3 b_5}{38416} - \frac{b_4 b_5^2}{1372} + \frac{b_3 b_5 b_6}{1372}\right) R_1 + \left(\frac{b_3}{14}\right)^7 = 0 & (12) \\ r_2 = -2 r_1^2 \cdot \frac{R_1 \left(5488 b_3 b_4^2 - 1372 b_3^2 b_5 + 19208 b_5 b_6\right) + A}{R_1 \left(-196 b_3^4 - 10976 b_4^3 + 2744 b_3 b_4 b_5 - 76832 b_6^2\right) + B} & (13) \\ r_4 = \frac{-b_3}{14 r_1 r_2} & (14) \\ A = b_3^6 - 28 b_3^2 b_4^3 + 28 b_3^3 b_4 b_5 - 196 b_4^2 b_5^2 + 98 b_3 b_5^3 + 7 b_3^4 b_6 + 196 b_3 b_4 b_5 b_6 & \square \\ B = b_3^5 b_4 - 28 b_3^2 b_4^2 b_5 + 14 b_3^3 b_5^2 + 28 b_3^3 b_4 b_6 & \square \end{cases}$$

「$b_3 = b_4 = b_5 = b_6 = 0$, $b_7 \neq 0$ の時」に r_1, r_2, r_4 を求める公式

$$r_1 = \sqrt[7]{-b_7}, \quad r_2 = r_4 = 0$$

上のいずれの場合も r_1^7, r_2^7, r_4^7 は下の「$R = 0$」の解となる.（↓評価してください）

In[3529]:=

```
R[{b3_, b4_, b5_, b6_, b7_}] =
  x^3 + (b3 b4/14 + b7) x^2 + (-b3^2 b4^2/9604 - b3^3 b5/38416 - b4 b5^2/1372 + b3 b5 b6/1372) x + (b3/14)^7;
```

【注】「$b_7 = 0$ のときは $r_1 = 0$」となり「$r_2 \sim r_6$ を r_1 で表す」という目的に使えません.
しかし，この時はτによって変換すると「$f_1 \leftrightarrow f_2$, $g_1 \leftrightarrow g_2$」となるから，この新しい g_1
で考えると（fが既約である事から）「$b_3 = b_4 = b_5 = b_6 = b_7 = 0$」となることはないです.

r_3, r_5, r_6 を r_1 で表す式 (「Formula2.nb」参照)

(b_i, c_i, d_i^+, d_i^-, e_i は, それぞれ g_1, g_2,
$f^+ = f_1 + f_2$, $f^- = \sqrt{-7}\,(f_1 - f_2)$, $\sqrt{-7}\, f_3$ の $(7 - i)$ 次の係数)

r_1, r_2, r_4 と r_3, r_5, r_6 の関係

$$\begin{pmatrix} r1 & r2 & r4 \\ r4^2 & r1^2 & r2^2 \\ r2^2\, r4 & r1\, r4^2 & r1^2\, r2 \\ r2^4 & r4^4 & r1^4 \end{pmatrix} \begin{pmatrix} r6 \\ r5 \\ r3 \end{pmatrix} = \begin{pmatrix} k1 \\ k2 \\ k3 \\ k4 \end{pmatrix},$$

但し
$$\begin{cases} k1 = -7\, a_2 & (16) \\ k2 = -\frac{1}{14}\, b_3 + \frac{21}{4}\, d_3^+ - \frac{7}{4}\, d_3^- - 14\, a_3 & (17) \\ k3 = \frac{1}{21}\, c_4 - \frac{1}{42}\, b_4 + \frac{49}{12}\, d_4^+ + \frac{49}{12}\, d_4^- - \frac{98}{3}\, a_4 & (18) \\ k4 = 7\, e_5 + \frac{1}{7}\, c_5 - \frac{3}{14}\, b_5 - \frac{343\,(d_5^+ + d_5^-)}{4} + b_3\, a_2 - 343\, a_5 & (19) \end{cases}$$

上の4×3行列をmatとすると,det1,det2,det3の少なくとも一つは0でないので, r_3, r_5, r_6 も r_1の式で表せます.

$$\begin{cases} det1 = det\,(16,\ 17,\ 18) = -\frac{1}{7}\, b_6 - \frac{1}{49}\, b_3^2 & \square \\ det2 = det\,(16,\ 17,\ 19) = -b_7 - \frac{4}{49}\, b_3\, b_4 & \square \\ det3 = det\,(16,\ 18,\ 19) = \frac{1}{49}\, b_4^2 - \frac{3}{196}\, b_3\, b_5 & \square \end{cases}$$

$$\begin{cases} b_3 = \cdots = b_6 = 0,\ b_7 \neq 0 \text{ のとき} & det2 = det\,(16,\ 17,\ 19) \neq 0 \\ b_3 = 0,\ b_4\, b_5\, b_6\, b_7 \neq 0 \text{ のとき} & det1 = det\,(16,\ 17,\ 18) \neq 0 \\ b_3 \neq 0 \text{ のとき} & det1 = det\,(16,\ 17,\ 18) \neq 0 \text{ または} det3 = det\,(16,\ 18,\ 19) \neq 0 \end{cases}$$

§1 例を2つ

「Formula1.nb」であげた例と同じですが，今回は厳密解を求めて公式の正しさを確認します.

1. $f = x^7 - 2x^5 + x^4 + 4x^3 - x^2 - 4x + 3$ (Gal = F_{42}, $b_3 \neq 0$)

まずf1,f2, g1,g2, b_i, c_i, d_i^+,d_i^- を求めます. ({f1, f2, g1, g2}, f^+=fp, f^- = fm の6式を表示しています)

In[3530]:=

```
f = x^7 - 2 x^5 + x^4 + 4 x^3 - x^2 - 4 x + 3 ;
{f1, f2, g1, g2} = findF12G12[f]
{a2, a3, a4, a5, a6, a7} = CoefficientList[f, x, 6] // Reverse;
{b3, b4, b5, b6, b7} = CoefficientList[g1, x, 5] // Reverse;
{c3, c4, c5, c6, c7} = CoefficientList[g2, x, 5] // Reverse;
Fp = f1 + f2 // Collect[#, x] &(*Fp=F⁺*)
Fm = Sqrt[-7] (f1 - f2) // Collect[#, x] &(*Fm=F⁻*)
{dp2, dp3, dp4, dp5, dp6, dp7} = CoefficientList[Fp, x, 6] // Reverse;
{dm2, dm3, dm4, dm5, dm6, dm7} = CoefficientList[Fm, x, 6] // Reverse;
```

Out[3531]=

$$\left\{-\frac{3}{2} - \frac{3 i \sqrt{91}}{2} - \frac{23 x}{2} + \frac{1}{2} i \sqrt{91}\, x - 17 x^2 + i \sqrt{91}\, x^2 + \frac{3 x^3}{2} - \frac{3}{2} i \sqrt{91}\, x^3 + 3 x^4 - i \sqrt{91}\, x^4 - 4 x^5 + x^7,\ -\frac{3}{2} + \frac{3 i \sqrt{91}}{2} - \frac{23 x}{2} - \frac{1}{2} i \sqrt{91}\, x - 17 x^2 - i \sqrt{91}\, x^2 + \frac{3 x^3}{2} + \frac{3}{2} i \sqrt{91}\, x^3 + 3 x^4 + i \sqrt{91}\, x^4 - 4 x^5 + x^7,\right.$$
$$\frac{381759}{2} - \frac{108045 \sqrt{13}}{2} - \frac{220549 x}{2} + \frac{64827 \sqrt{13}\, x}{2} + 20580 x^2 - 6174 \sqrt{13}\, x^2 - \frac{2401 x^3}{2} + \frac{1029 \sqrt{13}\, x^3}{2} + 196 x^4 + x^7,\ \frac{381759}{2} + \frac{108045 \sqrt{13}}{2} - \frac{220549 x}{2} -$$
$$\left.\frac{64827 \sqrt{13}\, x}{2} + 20580 x^2 + 6174 \sqrt{13}\, x^2 - \frac{2401 x^3}{2} - \frac{1029 \sqrt{13}\, x^3}{2} + 196 x^4 + x^7\right\}$$

Out[3535]=

$$-3 - 23 x - 34 x^2 + 3 x^3 + 6 x^4 - 8 x^5 + 2 x^7$$

Out[3536]=

$$21 \sqrt{13} - 7 \sqrt{13}\, x - 14 \sqrt{13}\, x^2 + 21 \sqrt{13}\, x^3 + 14 \sqrt{13}\, x^4$$

次に R(=eq)とその解solを求めます

In[3539]:=

```
eq = R[{b3, b4, b5, b6, b7}] // Simplify;
sol = x /. Solve[eq == 0, x, Cubics → True];
```

「r_1 = sol⟦1⟧ ^ (1 / 7)」としてr_1, r_2 と r_4 の厳密解を求めます. (r_1 のみ表示しています)

In[3541]:=

```
findr2[r1_] :=
 Simplify[-2 (r1) ^2 * ((r1^7 (5488 b3 b4^2 - 1372 b3^2 b5 + 19 208 b5 b6) + b3^6 - 28 b3^2 b4^3 +
       28 b3^3 b4 b5 - 196 b4^2 b5^2 + 98 b3 b5^3 + 7 b3^4 b6 + 196 b3 b4 b5 b6) /
     (r1^7 (-196 b3^4 - 10 976 b4^3 + 2744 b3 b4 b5 - 76 832 b6^2) + b3^5 b4 -
       28 b3^2 b4^2 b5 + 14 b3^3 b5^2 + 28 b3^3 b4 b6))]
```

In[3542]:=

```
r1 = sol[[1]]^(1/7)
r2 = findr2[r1];
r4 = -b3 / (14 r1 r2);
```

Out[3542]=

$$\left(\frac{2401}{6}\left(-145+39\sqrt{13}\right)-\left(1372\times 7^{2/3}\left(-116506+32289\sqrt{13}\right)\right)\Big/\left(3\left(\frac{1}{2}\left(-43982153605+12198286674\sqrt{13}+3\,i\sqrt{3\left(554388682885501321-153759755291383332\sqrt{13}\right)}\right)\right)^{1/3}\right)+\frac{49}{3}\left(\frac{7}{2}\left(-43982153605+12198286674\sqrt{13}+3\,i\sqrt{3\left(554388682885501321-153759755291383332\sqrt{13}\right)}\right)\right)^{1/3}\right)^{1/7}$$

「$b_3 \neq 0$」なので「det1 ≠ 0」または「det3 ≠ 0」のはずです．チェックします．

In[3545]:=

```
det1 = -1/7 b6 - 1/49 b3^2 // Simplify
```

Out[3545]=

$$-\frac{49}{2}\left(-611+189\sqrt{13}\right)$$

「det1≠0」なので　(16), (17), (18) は一次独立です．よって mat[[{1,2,3}]]とk1,k2,k3 から, r_3,r_5,r_6を求められます．(クラメールの公式を使います.)

In[3546]:=

```
rightHand = {k1, k2, k3} = {-7 a2, -1/14 b3 + 21/4 dp3 - 7/4 dm3 - 14 a3,
    1/21 c4 - 1/42 b4 + 49/12 (dp4 + dm4) - 98/3 a4} // Simplify;
mat1 = mat2 = mat3 = {{r1, r2, r4}, {r4^2, r1^2, r2^2 }, {r4 r2^2, r1 r4^2, r2 r1^2}};
mat1[[All, 1]] = mat2[[All, 2]] = mat3[[All, 3]] = rightHand;
{r6, r5, r3} = {Det[mat1], Det[mat2], Det[mat3]} / det1;
```

Lagrange分解式の逆変換を使って, $x_0 \sim x_6$（= Xs）を求めます．（ζは1の原始7乗根です.）

In[3550]:=

```
Xs = {x0, x1, x2, x3, x4, x5, x6} = 1 / 7
    {{1, 1, 1, 1, 1, 1}, {ξ^6, ξ^5, ξ^4, ξ^3, ξ^2, ξ}, {ξ^5, ξ^3, ξ, ξ^6, ξ^4, ξ^2},
      {ξ^4, ξ, ξ^5, ξ^2, ξ^6, ξ^3}, {ξ^3, ξ^6, ξ^2, ξ^5, ξ, ξ^4}, {ξ^2, ξ^4,
       ξ^6, ξ, ξ^3, ξ^5}, {ξ, ξ^2, ξ^3, ξ^4, ξ^5, ξ^6}}.{r1, r2, r3, r4, r5, r6};
```

長いので, x0のみ「Shortで」表示します.(通常表示では**数ページ**に渡ります.)

In[3551]:=

```
x0 // Short
```

Out[3551]//Short=

$$\frac{1}{7}\left(\left(\frac{2401}{6}\left(-145+39\sqrt{13}\right)-\frac{\ll 1\gg}{3\left(\ll 1\gg\right)^{1/3}}+\frac{49}{3}\left(\frac{7}{2}\left(\ll 1\gg\right)\right)^{1/3}\right)^{1/7}+\ll 9\gg\right)$$

この近似値を求めて, NSolveの数値解と比較します. 御覧の様に小数点5桁まで一致します.

In[3552]:=

```
N[Xs /. {ξ → Cos[2 Pi / 7] + I Sin[2 Pi / 7]}] // Sort
```

Out[3552]=

$\{-1.3888-1.12271\times 10^{-6}\,i,\ -1.0395+0.82652\,i,\ -1.0395-0.82652\,i,\ 0.736558+0.408262\,i,$
$0.736559-0.408261\,i,\ 0.997338+0.855759\,i,\ 0.997338-0.855758\,i\}$

In[3553]:=

```
x /. NSolve[f == 0, x]
```

Out[3553]=

$\{-1.3888,\ -1.0395-0.82652\,i,\ -1.0395+0.82652\,i,\ 0.736558-0.408262\,i,$
$0.736558+0.408262\,i,\ 0.997338-0.855759\,i,\ 0.997338+0.855759\,i\}$

【コメント】「Formula2.nb」で述べた様に, 「$r_1 \to r_1\,\xi$」としても「$x_0 \sim x_6$」の値は変わりません.

解の簡略化

上のx0の表現はいくらなんでも長すぎます. 原論文では理論的な事に注力していましたが, 私はできるだけ簡単な表現が欲しいです. そのためには $r_2{}^7$,$r_4{}^7 r_3{}^7$,$r_5{}^7$,$r_6{}^7$も「R=0」の解として求めて, 近似値を使いその7乗根を選びます. またMathematicaの3次方程式の解の公式は「分母の有理化」が行われていません. (「x^3+p x+q=0」の解を「x=u+v」とおくとき, u^3, v^3 は2次方程式「t^2+q t- p^3/27=0」の解です. u,vはその2解の3乗根をとる必要があり, その際「u v=-p/3」を満たすように偏角を選ばないといけませんが, 恐らくMathematicaは「v=-p/(3u)」としています.) 従って, 分母に3乗根を含まない 自作の解の公式 *solveCube* も使います.

In[3554]:=

```
solveCube[f_] := Module[{delta, g, u, v, power},
  delta = Coefficient[f, x, 2];
  g = Expand[f /. {x → x - delta / 3}];
  {q, p} = CoefficientList[g, x, 2];
  {u, v} = {(-q / 2 + Sqrt[q^2 / 4 + p^3 / 27]) ^ (1 / 3),
     (-q / 2 - Sqrt[q^2 / 4 + p^3 / 27]) ^ (1 / 3)};
  ω = Cos[2 / 3 Pi] + I Sin[2 / 3 Pi];
  power = Mod[Round[N[(Arg[-p / 3] - Arg[u] - Arg[ v])] / (2 Pi / 3)], 3];
  v = v * ω^power; (*Clear[ω];*)
  Simplify[{u + v, u ω + v ω^2, u ω^2 + v ω} - delta / 3]]
```

[1][r_1~ r_6 の近似値を求める. 求めた数値解は$r1~$r6に入れる]

In[3555]:=

```
sol = solveCube[eq];
r1 = sol⟦1⟧ ^ (1 / 7);
$r1 = N[r1];
$r2 = findr2[$r1];
$r4 = - b3 / (14 $r1 $r2);
det1 = - 1/7 b6 - 1/49 b3^2;
$mat1 = $mat2 = $mat3 =
    {{$r1, $r2, $r4}, {$r4^2, $r1^2, $r2^2 }, {$r4 $r2^2, $r1 $r4^2, $r2 $r1^2}};
rightHand = {- 7 a2, - 1/14 b3 + 21/4 dp3 - 7/4 dm3 - 14 a3, 1/21 c4 - 1/42 b4 + 49/12 (dp4 + dm4) - 98/3 a4};
$mat1⟦All, 1⟧ = $mat2⟦All, 2⟧ = $mat3⟦All, 3⟧ = rightHand;
{$r6, $r5, $r3} = {Det[$mat1], Det[$mat2], Det[$mat3]} / det1;
```

[2][r_1, r_2, r_4 の厳密値を求める. 求めた厳密値はr1,r2,r4に入れる]

In[3565]:=

```
posR2 = PositionSmallest[Abs[N[$r2^7 - sol⟦{2, 3}⟧]]]⟦1⟧ + 1;
(*R2=r2^7のsolにおけるposition*)
posR4 = 5 - posR2; (*R3=r3^7のsolにおけるposition.posR2=2→posR3=3,posR2=3→posR3=2*)
{R2, R4} = sol⟦{posR2, posR4}⟧;
{r2, r4} = {R2, R4} ^ (1 / 7);
power1 = Mod[Round[(N[{Arg[$r2] - Arg[r2], Arg[$r4] - Arg[r4]}]) / (2 Pi / 7)], 7];
{r2, r4} = {r2, r4} ξ^power1;
```

[3][r_3, r_5, r_6 の厳密値を求める. 求めた厳密値はr3,r5,r6に入れる]

Rの式で「$b_i \to c_i$」に変えた式をSとすると, $r_3{}^7$, $r_5{}^7$, $r_6{}^7$ は「S = 0」の解です. r_2, r_4 と同様にして r_3, r_5, r_6 を求めます.

In[3571]:=

```
sol2 = solveCube[R[{c3, c4, c5, c6, c7}]];
posR3 = PositionSmallest[Abs[$r3^7 - sol2]]⟦1⟧; (*R3=r3^7のsol2におけるposition*)
posR5 = PositionSmallest[Abs[$r5^7 - sol2]]⟦1⟧; (*R5=r5^7のsol2におけるposition*)
posR6 = 6 - (posR3 + posR5);
{R3, R5, R6} = sol2⟦{posR3, posR5, posR6}⟧;
{r3, r5, r6} = {R3, R5, R6} ^ (1 / 7);
power2 = Mod[Round[
    (N[{Arg[$r3] - Arg[r3], Arg[$r5] - Arg[r5], Arg[$r6] - Arg[r6] }]) / (2 Pi / 7)], 7];
{r3, r5, r6} = {r3, r5, r6} ξ^power2;
```

[4][r_k の値を使ってx_0~x_6 を求める]

In[3579]:=

```
X = {x0, x1, x2, x3, x4, x5, x6} = 1 / 7
    {{1, 1, 1, 1, 1, 1}, {ξ^6, ξ^5, ξ^4, ξ^3, ξ^2, ξ}, {ξ^5, ξ^3, ξ, ξ^6, ξ^4, ξ^2},
      {ξ^4, ξ, ξ^5, ξ^2, ξ^6, ξ^3}, {ξ^3, ξ^6, ξ^2, ξ^5, ξ, ξ^4}, {ξ^2, ξ^4,
       ξ^6, ξ, ξ^3, ξ^5}, {ξ, ξ^2, ξ^3, ξ^4, ξ^5, ξ^6}}.{r1, r2, r3, r4, r5, r6};
```

x_0 のみ今度は **省略なしで** 表示してみます．
ず〜と短くなりました． 但し「$\zeta = \cos(2\pi/7) + \iota \sin(2\pi/7)$」です．

In[3580]:=

```
x0 // Simplify
```

Out[3580]=

$$\frac{1}{2^{2/7}\times 3^{1/7}\times 7^{5/7}}\Bigg(\Big(2\Big(-7105+1911\sqrt{13}+2^{2/3}\Big(7\Big(-43982153605+12198286674\sqrt{13}-3\,\mathrm{i}\sqrt{39\left(42645283298884717-11827673483952564\sqrt{13}\right)}\Big)\Big)^{1/3}+2^{2/3}\Big(7\Big(-43982153605+12198286674\sqrt{13}+3\,\mathrm{i}\sqrt{39\left(42645283298884717-11827673483952564\sqrt{13}\right)}\Big)\Big)^{1/3}\Big)\Big)^{1/7}+$$

$$\Big(98\left(-145+39\sqrt{13}\right)+\mathrm{i}\,2^{2/3}\left(\mathrm{i}+\sqrt{3}\right)\Big(7\Big(-43982153605+12198286674\sqrt{13}-3\,\mathrm{i}\sqrt{39\left(42645283298884717-11827673483952564\sqrt{13}\right)}\Big)\Big)^{1/3}+2^{2/3}\left(-1-\mathrm{i}\sqrt{3}\right)\Big(7\Big(-43982153605+12198286674\sqrt{13}+3\,\mathrm{i}\sqrt{39\left(42645283298884717-11827673483952564\sqrt{13}\right)}\Big)\Big)^{1/3}\Big)^{1/7}+\Big(98\left(-145+39\sqrt{13}\right)+2^{2/3}\left(-1-\mathrm{i}\sqrt{3}\right)\Big(7\Big(-43982153605+12198286674\sqrt{13}-3\,\mathrm{i}\sqrt{39\left(42645283298884717-11827673483952564\sqrt{13}\right)}\Big)\Big)^{1/3}+\mathrm{i}\,2^{2/3}\left(\mathrm{i}+\sqrt{3}\right)\Big(7\Big(-43982153605+12198286674\sqrt{13}+3\,\mathrm{i}\sqrt{39\left(42645283298884717-11827673483952564\sqrt{13}\right)}\Big)\Big)^{1/3}\Big)^{1/7}\zeta^3+$$

$$\Big(2\Big(-49\left(145+39\sqrt{13}\right)+2^{2/3}\Big(7\Big(-43982153605-12198286674\sqrt{13}-3\,\mathrm{i}\sqrt{39\left(42645283298884717+11827673483952564\sqrt{13}\right)}\Big)\Big)^{1/3}+2^{2/3}\Big(7\Big(-43982153605-12198286674\sqrt{13}+3\,\mathrm{i}\sqrt{39\left(42645283298884717+11827673483952564\sqrt{13}\right)}\Big)\Big)^{1/3}\Big)\Big)^{1/7}\zeta^3+\Big(-98\left(145+39\sqrt{13}\right)+\mathrm{i}\,2^{2/3}\left(\mathrm{i}+\sqrt{3}\right)\Big(7\Big(-43982153605-12198286674\sqrt{13}-3\,\mathrm{i}\sqrt{39\left(42645283298884717+11827673483952564\sqrt{13}\right)}\Big)\Big)^{1/3}+2^{2/3}\left(-1-\mathrm{i}\sqrt{3}\right)\Big(7\Big(-43982153605-12198286674\sqrt{13}+3\,\mathrm{i}\sqrt{39\left(42645283298884717+11827673483952564\sqrt{13}\right)}\Big)\Big)^{1/3}\Big)^{1/7}\zeta^3+\Big(-98\left(145+39\sqrt{13}\right)+2^{2/3}\left(-1-\mathrm{i}\sqrt{3}\right)\Big(7\Big(-43982153605-12198286674\sqrt{13}-3\,\mathrm{i}\sqrt{39\left(42645283298884717+11827673483952564\sqrt{13}\right)}\Big)\Big)^{1/3}+\mathrm{i}\,2^{2/3}\left(\mathrm{i}+\sqrt{3}\right)\Big(7\Big(-43982153605-12198286674\sqrt{13}+3\,\mathrm{i}\sqrt{39\left(42645283298884717+11827673483952564\sqrt{13}\right)}\Big)\Big)^{1/3}\Big)^{1/7}\zeta^3\Bigg)$$

解の近似値を求めて， NSolveの数値解と比較します． 御覧の様に小数点5桁まで一致します．

In[3581]:=

```
N[X /. {ζ → Cos[2 Pi / 7] + I Sin[2 Pi / 7]}] // Chop // Sort
```

Out[3581]=

$\{-1.3888, -1.0395+0.82652\,\mathrm{i}, -1.0395-0.82652\,\mathrm{i}, 0.736558+0.408262\,\mathrm{i}, 0.736558-0.408262\,\mathrm{i}, 0.997338+0.855759\,\mathrm{i}, 0.997338-0.855759\,\mathrm{i}\}$

In[3582]:=

```
x /. NSolve[f == 0, x]
```

Out[3582]=

$\{-1.3888, -1.0395-0.82652\,\mathrm{i}, -1.0395+0.82652\,\mathrm{i}, 0.736558-0.408262\,\mathrm{i}, 0.736558+0.408262\,\mathrm{i}, 0.997338-0.855759\,\mathrm{i}, 0.997338+0.855759\,\mathrm{i}\}$

2. $f = x^7 - 7x^5 + 14x^3 - 7x + 3$ (Gal = F_{42}, $b_3 = \cdots = b_6 = 0$, $b_7 \neq 0$)

まずf1,f2, g1,g2, b_i, c_i, d_i^+,d_i^- を求めます. ({f1, f2, g1, g2}, f^+=fp, f^- = fmの6式を表示しています)

In[3583]:=

```
f = x^7 - 7 x^5 + 14 x^3 - 7 x + 3;
{f1, f2, g1, g2} = findF12G12[f];
{a2, a3, a4, a5, a6, a7} = CoefficientList[f, x, 6] // Reverse
{b3, b4, b5, b6, b7} = CoefficientList[g1, x, 5] // Reverse
{c3, c4, c5, c6, c7} = CoefficientList[g2, x, 5] // Reverse
Fp = f1 + f2 // Collect[#, x] &(*Fp=F⁺*)
Fm = Sqrt[-7] (f1 - f2) // Collect[#, x] &;(*Fm=F⁻*)
{dp2, dp3, dp4, dp5, dp6, dp7} = CoefficientList[Fp, x, 6] // Reverse
{dm2, dm3, dm4, dm5, dm6, dm7} = CoefficientList[Fm, x, 6] // Reverse;
f3 = findF3[f]
e5 = Sqrt[-7] Coefficient[f3, x, 2];
```

Out[3585]=

$\{-7, 0, 14, 0, -7, 3\}$

Out[3586]=

$\left\{0, 0, 0, 0, \frac{2470629}{2} + \frac{823543\sqrt{5}}{2}\right\}$

Out[3587]=

$\left\{0, 0, 0, 0, \frac{2470629}{2} - \frac{823543\sqrt{5}}{2}\right\}$

Out[3588]=

$39 - 112x + 112x^3 - 28x^5 + 2x^7$

Out[3590]=

$\{-28, 0, 112, 0, -112, 39\}$

Out[3592]=

$343 i \sqrt{35} - 2401x + 686x^3 - 49x^5 + x^7$

「$b_3 = b_4 = b_5 = b_6 = 0$」なので (13), (14) の式は使えず「$r_2 = r_4 = 0$」です.

このとき「r_1=-$\sqrt[7]{b_7}$」ですから,

In[3594]:=

```
r2 = r4 = 0;
r1 = -b7^(1/7)
```

Out[3595]=

$-\left(\frac{2470629}{2} + \frac{823543\sqrt{5}}{2}\right)^{1/7}$

「$b_3 = b_4 = 0$, $b_7 \neq 0$」なので「det2 = det (16, 17, 19) ≠ 0」です.
故に {k1, k2, k4} と mat⟦{1, 2, 4}⟧ を求め r_3, r_5, r_6 を求めます.

In[3596]:=

```
det2 = -b7 - 4/49 b3 b4;
rightHand = {-7 a2, -1/14 b3 + 21/4 dp3 - 7/4 dm3 - 14 a3,
   7 e5 + 1/7 c5 - 3/14 b5 - 343 (dp5 + dm5)/4 + b3 a2 - 343 a5} // Simplify
mat1 = mat2 = mat3 = {{r1, r2, r4}, {r4^2, r1^2, r2^2 }, {r2^4, r4^4, r1^4}};
mat1[[All, 1]] = mat2[[All, 2]] = mat3[[All, 3]] = rightHand;
{r6, r5, r3} = {Det[mat1], Det[mat2], Det[mat3]} / det2;
```

Out[3597]=

$$\{49, 0, 0\}$$

$x_0 \sim x_6$ を表示します. 但し「$\zeta = \cos(2\pi/7) + \iota \sin(2\pi/7)$」です.

In[3601]:=

```
X = {x0, x1, x2, x3, x4, x5, x6} =
  1 / 7 {{1, 1, 1, 1, 1, 1}, {ζ^6, ζ^5, ζ^4, ζ^3, ζ^2, ζ}, {ζ^5, ζ^3, ζ, ζ^6, ζ^4, ζ^2},
       {ζ^4, ζ, ζ^5, ζ^2, ζ^6, ζ^3}, {ζ^3, ζ^6, ζ^2, ζ^5, ζ, ζ^4},
       {ζ^2, ζ^4, ζ^6, ζ, ζ^3, ζ^5}, {ζ, ζ^2, ζ^3, ζ^4, ζ^5, ζ^6}}.
      {r1, r2, r3, r4, r5, r6} // Simplify // Chop // Sort
```

Out[3601]=

$$\Bigg\{-\left(\frac{2}{3+\sqrt{5}}\right)^{1/7} - \left(\frac{1}{2}\left(3+\sqrt{5}\right)\right)^{1/7},$$
$$-\left(\frac{1}{2}\left(3+\sqrt{5}\right)\right)^{1/7}\zeta^3 - \left(\frac{2}{3+\sqrt{5}}\right)^{1/7}\zeta^4,\ -\left(\frac{2}{3+\sqrt{5}}\right)^{1/7}\zeta^3 - \left(\frac{1}{2}\left(3+\sqrt{5}\right)\right)^{1/7}\zeta^4,$$
$$-\left(\frac{1}{2}\left(3+\sqrt{5}\right)\right)^{1/7}\zeta^2 - \left(\frac{2}{3+\sqrt{5}}\right)^{1/7}\zeta^5,\ -\left(\frac{2}{3+\sqrt{5}}\right)^{1/7}\zeta^2 - \left(\frac{1}{2}\left(3+\sqrt{5}\right)\right)^{1/7}\zeta^5,$$
$$-\left(\frac{1}{2}\left(3+\sqrt{5}\right)\right)^{1/7}\zeta - \left(\frac{2}{3+\sqrt{5}}\right)^{1/7}\zeta^6,\ -\left(\frac{2}{3+\sqrt{5}}\right)^{1/7}\zeta - \left(\frac{1}{2}\left(3+\sqrt{5}\right)\right)^{1/7}\zeta^6\Bigg\}$$

この場合は, 公式通りに r_1~r_6 を求めても解は簡単です.
次に 近似値を求めて, NSolveの数値解と比較します. 御覧の様に一致します.

In[3602]:=

```
X /. {ζ → Cos[2 Pi / 7] + I Sin[2 Pi / 7]} // N // Sort
```

Out[3602]=

```
{-2.01893, -1.25878 + 0.215665 ⅈ, -1.25878 - 0.215665 ⅈ, 0.449255 + 0.268929 ⅈ,
 0.449255 - 0.268929 ⅈ, 1.819 - 0.119685 ⅈ, 1.819 + 0.119685 ⅈ}
```

In[3603]:=

```
x /. NSolve[f == 0, x]
```

Out[3603]=

```
{-2.01893, -1.25878 - 0.215665 ⅈ, -1.25878 + 0.215665 ⅈ, 0.449255 - 0.268929 ⅈ,
 0.449255 + 0.268929 ⅈ, 1.819 - 0.119685 ⅈ, 1.819 + 0.119685 ⅈ}
```